

CALCUL DES VARIATIONS : EXERCICES CORRIGÉS

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Exercice M1

Enoncé

Déterminer les extrêmes éventuelles des fonctionnelles suivantes :

1. $V(x) = \int_0^1 (tx + 2x^2) dt$ avec $x(0) = 1$ et $x(1) = 2$.

2. $V(x) = \int_0^1 (2xe^t + x^2 + x') dt$ avec $x(0) = 2$ et $x(1) = 2e^2 + e^{-2}$.

3. $V(x) = \int_0^1 (tx' + x^2) dt$ avec $x(0) = 1$ et $x(1) = 0$.

4. $V(x) = \int_0^1 (x^2 + x'^2 + 2xe^t) dt$.

5. $V(x) = \int_1^2 \left(\frac{x^2}{t^2} \right) dt$ avec $x(1) = 1$ et $x(2) = 2$.

6. $V(x) = \int_0^{\frac{\pi}{2}} (x^2 - x'^2) dt$ avec $x(0) = 1$ et $x\left(\frac{\pi}{2}\right) = 0$.

7. $V(x) = \int_0^1 \sqrt{1 + x'^2} dt$.

8. $V(x) = \int_0^{\frac{\pi}{2}} (x^2 - x'^2) dt$ avec $x(0) = 0$ et $x\left(\frac{\pi}{2}\right) = 1$.

Quand une extrême existe, précisez sa nature.

Solution

Toutes les questions se traitent "mécaniquement". On commence par charger le paquetage **VariationalCalculus** pour faire appel aux commandes **EulerLagrange** et **Convex**.

```
> restart;  
with(VariationalCalculus):
```

Ceci fait, on donnera dans chaque cas d'espèce l'expression de l'intégrande; on récupèrera l'équation d'Euler-Lagrange qu'on résoudra avec **dsolve**; enfin, on étudiera la nature de l'extrême.

1. On pose $fI(t) = tx + 2x^2$.

```
> f1:=t*x(t)+2*diff(x(t),t)^2;
```

```

sol1:=EulerLagrange(f1,t,x(t));
eulag1:=sol1[1];
dsolve({eulag1,x(0)=1,x(1)=2},x(t));
Convex(f1,t,x(t));

```

$$f1 := tx(t) + 2 \left(\frac{d}{dt} x(t) \right)^2$$

$$sol1 := \left\{ t - 4 \left(\frac{d^2}{dt^2} x(t) \right) \right\}$$

$$eulag1 := t - 4 \left(\frac{d^2}{dt^2} x(t) \right)$$

$$x(t) = \frac{1}{24} t^3 + \frac{23}{24} t + 1$$

$$\begin{bmatrix} 0 & 0 \\ 0 & 4 \end{bmatrix}, true$$

(1.1)

Le programme admet une extrémale qui minimise globalement la fonctionnelle.

2. On pose $f2(t) = 2xe^t + x^2 + x'$

```

> f2:=2*x(t)*exp(t)+x(t)^2+diff(x(t),t)^2;
sol2:=EulerLagrange(f2,t,x(t));
eulag2:=sol2[1];
dsolve({eulag2,x(0)=2,x(1)=2*exp(2)+exp(-2)},x(t));
Convex(f2,t,x(t));

```

$$f2 := 2x(t)e^t + x(t)^2 + \left(\frac{d}{dt} x(t) \right)^2$$

$$sol2 := \left\{ 2e^t + 2x(t) - 2 \left(\frac{d^2}{dt^2} x(t) \right) \right\}$$

$$eulag2 := 2e^t + 2x(t) - 2 \left(\frac{d^2}{dt^2} x(t) \right)$$

$$x(t) = \frac{1}{2} \frac{e^t(4e^2 + 2e^{-2} - 4e^{-1} - e)}{e - e^{-1}} - \frac{1}{2} \frac{e^{-t}(-5e + 4e^2 + 2e^{-2})}{e - e^{-1}} + \frac{1}{2} te^t$$

$$\begin{bmatrix} 2 & 0 \\ 0 & 2 \end{bmatrix}, true$$

(1.2)

Le programme admet une extrémale qui minimise globalement la fonctionnelle.

3. On pose $f3(t) = tx'(t) + x'(t)^2$.

```

> f3:=t*diff(x(t),t)+diff(x(t),t)^2;
sol3:=EulerLagrange(f3,t,x(t));
eulag3:=sol3[1];
dsolve({eulag3,x(0)=1,x(1)=0},x(t));
Convex(f3,t,x(t));

```

$$\begin{aligned}
 f3 &:= t \left(\frac{d}{dt} x(t) \right) + \left(\frac{d}{dt} x(t) \right)^2 \\
 sol3 &:= \left\{ -1 - 2 \left(\frac{d^2}{dt^2} x(t) \right), t + 2 \left(\frac{d}{dt} x(t) \right) = K_1 \right\} \\
 eulag3 &:= -1 - 2 \left(\frac{d^2}{dt^2} x(t) \right) \\
 x(t) &= -\frac{1}{4} t^2 - \frac{3}{4} t + 1 \\
 &\left[\begin{array}{cc} 0 & 0 \\ 0 & 2 \end{array} \right], true
 \end{aligned}
 \tag{1.3}$$

Le programme admet une extrémale qui minimise globalement la fonctionnelle.

4. On pose $f4(t) = x(t)^2 + x'(t)^2 + 2x(t)e^t$.

```

> f4:=x(t)^2+diff(x(t),t)^2+2*x(t)*exp(t);
sol4:=EulerLagrange(f4,t,x(t));
eulag4:=sol4[1];
dsolve(eulag4,x(t));
Convex(f4,t,x(t));

```

$$\begin{aligned}
 f4 &:= 2x(t)e^t + x(t)^2 + \left(\frac{d}{dt} x(t) \right)^2 \\
 sol4 &:= \left\{ 2e^t + 2x(t) - 2 \left(\frac{d^2}{dt^2} x(t) \right) \right\} \\
 eulag4 &:= 2e^t + 2x(t) - 2 \left(\frac{d^2}{dt^2} x(t) \right) \\
 x(t) &= e^t C_2 + e^{-t} C_1 + \frac{1}{2} t e^t \\
 &\left[\begin{array}{cc} 2 & 0 \\ 0 & 2 \end{array} \right], true
 \end{aligned}
 \tag{1.4}$$

Pour tout jeu de deux conditions aux bornes, le programme admet une extrémale qui minimise globalement la fonctionnelle.

5. On pose $f5(t) = \frac{x'(t)^2}{t^2}$.

```

> f5:=diff(x(t),t)^2/t^2;
sol5:=EulerLagrange(f5,t,x(t));
eulag5:=sol5[1];
dsolve({eulag5,x(1)=1,x(2)=2},x(t));
Convex(f5,t,x(t));

```

$$f5 := \frac{\left(\frac{d}{dt} x(t) \right)^2}{t^2}$$

$$sol5 := \left\{ -\frac{2 \left(\frac{d^2}{dt^2} x(t) \right)}{t^2} + \frac{4 \left(\frac{d}{dt} x(t) \right)}{t^3}, \frac{2 \left(\frac{d}{dt} x(t) \right)}{t^2} = K_1 \right\}$$

$$eulag5 := -\frac{2 \left(\frac{d^2}{dt^2} x(t) \right)}{t^2} + \frac{4 \left(\frac{d}{dt} x(t) \right)}{t^3}$$

$$x(t) = \frac{6}{7} + \frac{1}{7} t^3$$

$$\begin{bmatrix} 0 & 0 \\ 0 & \frac{2}{t^2} \end{bmatrix}, 0 \leq \frac{2}{t^2}$$

(1.5)

Le programme admet une extrémale qui minimise globalement la fonctionnelle.

6. On pose $f6(t) = x(t)^2 - x'(t)^2$.

```
> f6:=x(t)^2-diff(x(t),t)^2;
sol6:=EulerLagrange(f6,t,x(t));
eulag6:=sol6[1];
dsolve({eulag6,x(0)=1,x(Pi/2)=0},x(t));
Convex(f6,t,x(t));
```

$$f6 := x(t)^2 - \left(\frac{d}{dt} x(t) \right)^2$$

$$sol6 := \left\{ 2x(t) + 2 \left(\frac{d^2}{dt^2} x(t) \right), x(t)^2 + \left(\frac{d}{dt} x(t) \right)^2 = K_1 \right\}$$

$$eulag6 := 2x(t) + 2 \left(\frac{d^2}{dt^2} x(t) \right)$$

$$x([t]) = \cos([t])$$

$$\begin{bmatrix} 2 & 0 \\ 0 & -2 \end{bmatrix}, false$$

(1.6)

Le programme admet une extrémale mais, puisque les valeurs propres de la matrice hessienne sont de signe opposé, elle ne correspond ni à un minimum ni à un maximum global de la fonctionnelle.

7. On pose $f7(t) = (1 + x'(t)^2)^{\frac{1}{2}}$.

```
> f7:=(1+diff(x(t),t)^2)^(1/2);
sol7:=EulerLagrange(f7,t,x(t));
eulag7:=sol7[1];
dsolve(eulag7,x(t));
Convex(f7,t,x(t));
```

$$f7 := \sqrt{1 + \left(\frac{d}{dt} x(t) \right)^2}$$

$$\begin{aligned}
sol7 := & \left\{ \frac{\left(\frac{d}{dt} x(t)\right)^2 \left(\frac{d^2}{dt^2} x(t)\right)}{\left(1 + \left(\frac{d}{dt} x(t)\right)^2\right)^{3/2}} - \frac{\frac{d^2}{dt^2} x(t)}{\sqrt{1 + \left(\frac{d}{dt} x(t)\right)^2}}, \frac{\frac{d}{dt} x(t)}{\sqrt{1 + \left(\frac{d}{dt} x(t)\right)^2}} = K_1, \right. \\
& \left. \sqrt{1 + \left(\frac{d}{dt} x(t)\right)^2} - \frac{\left(\frac{d}{dt} x(t)\right)^2}{\sqrt{1 + \left(\frac{d}{dt} x(t)\right)^2}} = K_2 \right\} \\
eulag7 := & \frac{\left(\frac{d}{dt} x(t)\right)^2 \left(\frac{d^2}{dt^2} x(t)\right)}{\left(1 + \left(\frac{d}{dt} x(t)\right)^2\right)^{3/2}} - \frac{\frac{d^2}{dt^2} x(t)}{\sqrt{1 + \left(\frac{d}{dt} x(t)\right)^2}} \\
& x(t) = _C1 t + _C2 \\
\left[\begin{array}{cc} 0 & 0 \\ 0 & -\frac{\left(\frac{d}{dt} x(t)\right)^2}{\left(1 + \left(\frac{d}{dt} x(t)\right)^2\right)^{3/2}} + \frac{1}{\sqrt{1 + \left(\frac{d}{dt} x(t)\right)^2}} \end{array} \right], 0 \leq \frac{1}{\left(1 + \left(\frac{d}{dt} x(t)\right)^2\right)^{3/2}} \quad (1.7)
\end{aligned}$$

Quelles que soient les conditions aux bornes, le programme admet une extrémale qui minimise globalement la fonctionnelle (les deux valeurs propres de la matrice hessienne sont positives ou nulles).

8. On pose $f8(t) = x(t)^2 - x'(t)^2$.

```

> f8:=x(t)^2-diff(x(t),t)^2;
sol8:=EulerLagrange(f8,t,x(t));
eulag8:=sol8[1];
dsolve({eulag8,x(0)=0,x(Pi/2)=1},x(t));
Convex(f8,t,x(t));

```

$$\begin{aligned}
f8 & := x(t)^2 - \left(\frac{d}{dt} x(t)\right)^2 \\
sol8 & := \left\{ 2x(t) + 2\left(\frac{d^2}{dt^2} x(t)\right), x(t)^2 + \left(\frac{d}{dt} x(t)\right)^2 = K_1 \right\} \\
eulag8 & := 2x(t) + 2\left(\frac{d^2}{dt^2} x(t)\right) \\
x([t]) & = \sin([t]) \\
\begin{bmatrix} 2 & 0 \\ 0 & -2 \end{bmatrix}, & \text{false} \quad (1.8)
\end{aligned}$$

Le programme admet une extrémale mais, puisque les valeurs propres de la matrice hessienne sont de signe opposé, elle ne correspond ni à un minimum ni à un maximum global de la fonctionnelle.

Exercice M2

Enoncé

Etudier le programme :

$$\text{ext } J = \int_0^T x'(t) dt \text{ avec } x(0) = \alpha \text{ et } x(T) = \beta$$

Solution

```
> restart;
with(VariationalCalculus):
j:=diff(x(t),t);
sol:=EulerLagrange(j,t,x(t));
sol1:=dsolve({sol[1],x(0)=alpha,x(T)=beta},x(t));
```

$$j := \frac{d}{dt} x(t)$$

$$\text{sol} := \{0, 0 = K_2, 1 = K_1\}$$

Error, (in dsolve) not an ODE system with respect to the unknowns [x(t)]

L'équation d'Euler-Lagrange est réduite à ... $0 = 0$. Autrement dit, n'importe quelle fonction $x(t)$ vérifiant les conditions aux bornes convient. Le programme admet donc une infinité d'extrêmes, qui sont à la fois des minima et des maxima globaux puisque l'intégrande est à la fois convexe et concave comme le montre la matrice hessienne.

```
> Convex(f,t,x(t));
```

$$\begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}, \text{true}$$

(2.1)

Exercice M3

Enoncé

Expliquez pourquoi le programme :

$$\text{ext } J = \int_0^5 (t + y(t)^2 + 3 y'(t)) dt \text{ avec } y(0) = 3 \text{ et } y(5) = 3$$

n'admet pas de solution.

Solution

Le traitement direct du problème amène un message d'erreur :

```
> restart;
with(VariationalCalculus):
h:=t+y(t)^2+3*diff(y(t),t);
sol:=EulerLagrange(h,t,y(t));
eulag:=sol[1];
sol_eulag:=dsolve({eulag,y(0)=3,y(5)=3},y(t));
```

$$h := t + y(t)^2 + 3 \left(\frac{d}{dt} y(t) \right)$$

$$\text{sol} := \{2 y(t)\}$$

```
eulag := 2 y(t)
```

```
Error, (in dsolve) expected an ordinary differential equation.
```

```
Received: 2*y(t)
```

L'équation d'Euler-Lagrange n'est pas une équation différentielle. En fait, l'extrémale est la fonction constante nulle : $y(t) = 0$. Cette fonction ne satisfait pas les conditions aux bornes, par conséquent le problème n'admet pas de solution même si l'intégrande est convexe.

```
> Convex(h,t,y(t));
```

$$\begin{bmatrix} 2 & 0 \\ 0 & 0 \end{bmatrix}, true \quad (3.1)$$

Exercice M4

Enoncé

On considère la fonctionnelle dépendant de deux variables $V(x, y) = \int_0^T (x + y + x^2 + y^2) dt$. Après avoir compulsé l'aide en ligne, donnez le système des équations d'Euler-Lagrange puis déterminez les extrémales $x^*(t)$ et $y^*(t)$.

Solution

Deux fonctions inconnues x et y interviennent dans la fonctionnelle. Ce n'est pas un problème pour Maple. L'aide en ligne indique que la commande **EulerLagrange** accepte en premier argument un intégrande dépendant de plusieurs fonctions inconnues. Le troisième argument est la liste ou l'ensemble composé des fonctions inconnues.

```
> restart;
```

```
with(VariationalCalculus):
```

```
f:=x(t)+y(t)+diff(x(t),t)^2+diff(y(t),t)^2;
```

```
sol:=EulerLagrange(f,t,[x(t),y(t)]);
```

```
eulag1:=sol[1];
```

```
eulag2:=sol[2];
```

$$f := x(t) + y(t) + \left(\frac{d}{dt} x(t) \right)^2 + \left(\frac{d}{dt} y(t) \right)^2$$

$$\text{sol} := \left\{ 1 - 2 \left(\frac{d^2}{dt^2} x(t) \right), 1 - 2 \left(\frac{d^2}{dt^2} y(t) \right), x(t) + y(t) - \left(\frac{d}{dt} x(t) \right)^2 - \left(\frac{d}{dt} y(t) \right)^2 \right. \\ \left. = K_1 \right\}$$

$$eulag1 := 1 - 2 \left(\frac{d^2}{dt^2} x(t) \right)$$

$$eulag2 := 1 - 2 \left(\frac{d^2}{dt^2} y(t) \right) \quad (4.1)$$

La commande **dsolve** résout aisément ce système de deux équations différentielles du second ordre.

```
> dsolve([eulag1,eulag2],[x(t),y(t)]);
```

$$\left\{ x(t) = \frac{1}{4} t^2 + _C1 t + _C2, y(t) = \frac{1}{4} t^2 + _C3 t + _C4 \right\} \quad (4.2)$$

Exercice M5

Enoncé

Soit le programme $\text{ext } J = \int_0^2 (t y^2 + y y' + y''^2) dt$ avec $x(0) = 0$ et $x(2) = 8$. Après avoir consulté

l'aide en ligne, cherchez l'expression générale de l'extrémale de J . Donnez une représentation graphique de quelques extrémales particulières.

Solution

La fonctionnelle dépend d'une fonction inconnue y mais l'intégrande combine cette fonction, sa dérivée première et sa dérivée seconde.

> **restart:**

with(VariationalCalculus):

f:=t*y(t)^2+y(t)*diff(y(t),t)+diff(diff(y(t),t),t)^2;

$$f := t y(t)^2 + y(t) \left(\frac{d}{dt} y(t) \right) + \left(\frac{d^2}{dt^2} y(t) \right)^2 \quad (5.1)$$

Sur la suggestion de l'aide en ligne, on effectue le changement de variable $y'(t) = x(t)$, d'où $y''(t) = x'(t)$.

> **g:=subs({diff(y(t),t)=x(t),diff(diff(y(t),t),t)=diff(x(t),t)},f);**

$$g := t y(t)^2 + y(t) x(t) + \left(\frac{d}{dt} x(t) \right)^2 \quad (5.2)$$

On applique la commande **EulerLagrange** à cet intégrande, qui dépend des fonctions x et y .

> **sol:=EulerLagrange(g,t,[x(t),y(t)]);**

eulag1:=sol[1];

eulag2:=sol[2];

$$\begin{aligned} \text{sol} &:= \left\{ 2 t y(t) + x(t), y(t) - 2 \left(\frac{d^2}{dt^2} x(t) \right) \right\} \\ \text{eulag1} &:= 2 t y(t) + x(t) \\ \text{eulag2} &:= y(t) - 2 \left(\frac{d^2}{dt^2} x(t) \right) \end{aligned} \quad (5.3)$$

On nomme **ystar** l'extrémale obtenue par résolution du système différentiel formé par **eulag1** et **eulag2**. Afin de préparer la représentation graphique, elle dépend non seulement de t mais aussi des deux constantes d'intégration $_C1$ et $_C2$.


> **dsolve([eulag1=0,eulag2=0],[x(t),y(t)];assign(%);ystar:=unapply(y(t),(t,_C1,_C2));**

$$\left\{ x(t) = -2 \sqrt{t} (_C1 \text{BesselJ}(1, \sqrt{t}) + _C2 \text{BesselY}(1, \sqrt{t})), y(t) \right\}$$

$$= \frac{C_1 \text{BesselJ}(1, \sqrt{t}) + C_2 \text{BesselY}(1, \sqrt{t})}{\sqrt{t}} \Bigg\} \\ ystar := (t, C_1, C_2) \rightarrow \frac{C_1 \text{BesselJ}(1, \sqrt{t}) + C_2 \text{BesselY}(1, \sqrt{t})}{\sqrt{t}} \quad (5.4)$$

Il ne reste plus qu'à donner des valeurs particulières aux constantes pour obtenir des trajectoires particulières.

```
> plot([ystar(t,1,1), ystar(t,1,2), ystar(t,2,1), ystar(t,2,2)], t=
0..100, y=-1..1);
```



▼ Exercice E1 (microdynamique du monopole)

Enoncé

Dans l'article de G.C. Evans "The Dynamics of the Monopoly" (American Mathematical Monthly, February 1924), un monopole produit et vend un bien unique en quantité Q . Sa fonction de coût de production est quadratique : $C = \alpha Q^2 + \beta Q + \gamma$ où α , β et γ sont des paramètres strictement positifs. La quantité produite dépend du prix et des variations de prix : $Q = a - bP + hP'$, avec $a > 0$, $b > 0$ et $h \neq 0$. Les questions suivantes permettent de saisir la substance du modèle de Evans.

1. Donner l'expression du profit **Profit** en fonction du prix et de ses variations.

2. L'objectif de la firme est de déterminer la trajectoire optimale du prix $P(t)$ qui maximise le profit sur la période $[0;T]$. Ecrire le programme intertemporel de la firme.
3. Déterminer l'extrémale du programme.
4. Représenter graphiquement l'extrémale correspondant aux valeurs de paramètres $\alpha = 0.05$, $\beta = 0.1$, $a = 100$, $b = 0.25$, $h = 1$, $T = 10$ et pour les conditions aux bornes $P(0) = P(10) = 150$.

Solution

1. Une fois posées les expressions du produit et du coût, il est facile d'en déduire le profit par différence entre le chiffre d'affaires et le coût.

```
> restart;
Prod:=a-b*P(t)+h*diff(P(t),t);
Cout:=alpha*Prod^2+beta*Prod+gamma;
Profit:=P(t)*Prod-Cout;
```

$$\begin{aligned}
 \text{Prod} &:= a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \\
 \text{Cout} &:= \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right)^2 + \beta \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) + \gamma \\
 \text{Profit} &:= P(t) \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) - \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right)^2 - \beta \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) - \gamma \quad (6.1)
 \end{aligned}$$

2. Le programme de la firme est la maximisation de la somme des flux de profit, soit :

$$\begin{aligned}
 &\max \\
 &\int_0^T \left(P(t) \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) - \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right)^2 - \beta \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) - \gamma \right) dt
 \end{aligned}$$

3. L'extrémale, qu'on note ici **Pstar**, est calculée par la commande **EulerLagrange** du paquetage **VariationalCalculus**.

```
> with(VariationalCalculus):
sol:=EulerLagrange(Profit,t,P(t));
eulag:=sol[1];
dsolve(eulag,P(t));
Pstar:=rhs(%);
```

$$\begin{aligned}
 \text{sol} &:= \left\{ a - 2 b P(t) + 2 \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) b + \beta b + 2 \alpha \left(-b \left(\frac{d}{dt} P(t) \right) + h \left(\frac{d^2}{dt^2} P(t) \right) \right) h, P(t) \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) - \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right)^2 - \beta \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) - \gamma - \left(\frac{d}{dt} P(t) \right) \left(P(t) h - 2 \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) h - \beta h \right) = K_1 \right\}
 \end{aligned}$$

$$\begin{aligned}
eulag &:= a - 2 b P(t) + 2 \alpha \left(a - b P(t) + h \left(\frac{d}{dt} P(t) \right) \right) b + \beta b + 2 \alpha \left(-b \left(\frac{d}{dt} P(t) \right) \right. \\
&\quad \left. + h \left(\frac{d^2}{dt^2} P(t) \right) \right) h \\
P(t) &= e^{\frac{\sqrt{b} \sqrt{1+\alpha b} t}{h\sqrt{\alpha}}} C_2 + e^{-\frac{\sqrt{b} \sqrt{1+\alpha b} t}{h\sqrt{\alpha}}} C_1 + \frac{1}{2} \frac{(2\alpha a + \beta) b + a}{b(1+\alpha b)} \\
Pstar &:= e^{\frac{\sqrt{b} \sqrt{1+\alpha b} t}{h\sqrt{\alpha}}} C_2 + e^{-\frac{\sqrt{b} \sqrt{1+\alpha b} t}{h\sqrt{\alpha}}} C_1 + \frac{1}{2} \frac{(2\alpha a + \beta) b + a}{b(1+\alpha b)} \tag{6.2}
\end{aligned}$$

4. Pour représenter graphiquement l'extrémale avec les valeurs fournies et les conditions aux bornes, on commence par récupérer l'expression de **eulag** avant de résoudre cette équation différentielle avec les conditions aux bornes. On nomme **Psimul** la trajectoire solution.


```
> alpha,beta,a,b,h:=0.05,0.1,100,0.25,1;
eulag;
```

$$\begin{aligned}
&\alpha, \beta, a, b, h := 0.05, 0.1, 100, 0.25, 1 \\
&102.5250 - 0.506250 P(t) + 0.10 \left(\frac{d^2}{dt^2} P(t) \right) \tag{6.3}
\end{aligned}$$

```
> dsolve({eulag,P(0)=150,P(10)=150},P(t));
Psimul:=unapply(rhs(%),t);
```

$$\begin{aligned}
P(t) &= \frac{1418}{27} \frac{e^{\frac{9}{4}t} \left(-1 + e^{-\frac{45}{2}} \right)}{e^{\frac{45}{2}} - e^{-\frac{45}{2}}} - \frac{1418}{27} \frac{e^{-\frac{9}{4}t} \left(e^{\frac{45}{2}} - 1 \right)}{e^{\frac{45}{2}} - e^{-\frac{45}{2}}} + \frac{5468}{27} \\
Psimul &:= t \rightarrow \frac{1418}{27} \frac{e^{\frac{9}{4}t} \left(-1 + e^{-\frac{45}{2}} \right)}{e^{\frac{45}{2}} - e^{-\frac{45}{2}}} - \frac{1418}{27} \frac{e^{-\frac{9}{4}t} \left(e^{\frac{45}{2}} - 1 \right)}{e^{\frac{45}{2}} - e^{-\frac{45}{2}}} + \frac{5468}{27} \tag{6.4}
\end{aligned}$$

```
> plot(Psimul(t),t=0..10);
```



▼ Exercice E2 (arbitrage inflation-chômage)

Enoncé

Cet exercice s'inspire du modèle célèbre de Dean Taylor développé dans l'article "Stopping Inflation in the Dornbusch Model : Optimal Monetary Policies with Alternate Price-Adjustment Equations" (Journal of Macroeconomics, 1989). L'idée générale est que l'inflation et le chômage sont deux fléaux qui génèrent des pertes sociales. Si un arbitrage à la Phillips existe entre les deux, que doit être la meilleure combinaison inflation-chômage dans le temps? Dans le modèle et l'exercice proposé ci-dessous, le chômage apparaît indirectement à travers la différence entre le niveau de production de plein emploi et le niveau de production courant.

On note λ la perte sociale générée par l'inflation et le chômage, Y la production courante, Y_p la production de plein emploi, P le taux d'inflation courant et P_e le taux d'inflation anticipé.

Le modèle d'arbitrage intertemporel entre l'inflation et le chômage repose sur les trois équations suivantes :

$$(1) \lambda = (Y - Y_p)^2 + \alpha P^2, \alpha > 0.$$

$$(2) P = -\beta (Y - Y_p) + P_e, \beta > 0.$$

$$(3) \frac{d}{dt} P_e = j (P - P_e), 0 < j \leq 1.$$

L'équation (1) signifie qu'à toute date la perte sociale est la somme du carré des écarts entre l'objectif

souhaitable (Pe pour la production, inflation nulle pour les prix) et la valeur courante. L'équation (2) est la relation de Phillips augmentée des anticipations. L'équation (3) fait l'hypothèse que les anticipations de prix sont adaptatives.

On veut répondre aux trois questions suivantes :

1. Montrer qu'à toute date la perte sociale est une fonction des anticipations de prix et de leurs variations.
2. Le problème des pouvoirs publics est de sélectionner la trajectoire optimale de Pe qui minimise la perte sociale sur l'intervalle temporel $[0;T]$ en tenant compte du fait que les agents ont une préférence pour le présent mesurée par le taux ρ et que, partant du taux d'inflation anticipé $Pe(0) = Pe0 > 0$, il faut atteindre l'objectif $Pe(T) = 0$. Ecrire le programme correspondant.
3. Déterminer l'extrémale du programme.
4. Représentez graphiquement l'extrémale en prenant $\rho = 0.05$, $\alpha = 1$, $\beta = 0.5$, $j = 0.2$ et les conditions aux bornes $Pe(0) = 15$ et $Pe(40) = 0$.

Solution

Avant tout, il faut poser les équations structurelles.

```
> restart;
eq1:=lambda=(Y(t)-Yp(t))^2+alpha*P(t)^2;
eq2:=P(t)=-beta*(Y(t)-Yp(t))+Pe(t);
eq3:=diff(Pe(t),t)=j*(P(t)-Pe(t));
```

$$eq1 := \lambda = (Y(t) - Yp(t))^2 + \alpha P(t)^2$$

$$eq2 := P(t) = -\beta (Y(t) - Yp(t)) + Pe(t)$$

$$eq3 := \frac{d}{dt} Pe(t) = j (P(t) - Pe(t)) \quad (7.1)$$

1. Comme souvent avec Maple, il faut de nombreuses manipulations pour obtenir un résultat qui prendrait peu de temps à la main. On propose les étapes suivantes pour montrer que la perte sociale instantanée ne dépend en définitive que des anticipations de prix et de leurs variations.

```
> eq31:=P(t)=solve(eq3,P(t));
```

$$eq31 := P(t) = \frac{\frac{d}{dt} Pe(t) + j Pe(t)}{j} \quad (7.2)$$

```
> eq21:=subs(Y(t)-Yp(t)=A,eq2);eq22:=A=solve(eq21,A);eq23:=subs
(eq31,eq22);
```

$$eq21 := P(t) = -\beta A + Pe(t)$$

$$eq22 := A = \frac{-P(t) + Pe(t)}{\beta}$$

$$eq23 := A = \frac{-\frac{\frac{d}{dt} Pe(t) + j Pe(t)}{j} + Pe(t)}{\beta} \quad (7.3)$$

```
> eq11:=subs(Y(t)-Yp(t)=A,eq1);eq12:=subs([eq23,eq31],eq11);
eq13:=lambda=simplify(rhs(eq12)) assuming j::real,beta::real,
alpha::real;
```

$$eq11 := \lambda = A^2 + \alpha P(t)^2$$

$$eq12 := \lambda = \frac{\left(-\frac{\frac{d}{dt} Pe(t) + j Pe(t)}{j} + Pe(t) \right)^2}{\beta^2} + \frac{\alpha \left(\frac{d}{dt} Pe(t) + j Pe(t) \right)^2}{j^2}$$

$$eq13 := \lambda \tag{7.4}$$

$$= \frac{\left(\frac{d}{dt} Pe(t) \right)^2 + \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right)^2 + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j Pe(t) + \alpha \beta^2 j^2 Pe(t)^2}{j^2 \beta^2}$$

2. Il s'agit de minimiser la somme actualisée des flux de perte sociale, donc le problème variationnel est le suivant :

$$\min \int_0^T e^{-\rho t} \left(\frac{\left(\frac{d}{dt} Pe(t) \right)^2 + \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right)^2 + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j Pe(t) + \alpha \beta^2 j^2 Pe(t)^2}{j^2 \beta^2} \right) dt$$

avec les conditions aux bornes $Pe(0) > 0$ et $Pe(T) = 0$.

On notera **f** l'intégrande.

$$> \mathbf{f := \exp(-\rho * t) * rhs(eq13);}$$

$$f := \tag{7.5}$$

$$\frac{1}{j^2 \beta^2} \left(e^{-\rho t} \left(\left(\frac{d}{dt} Pe(t) \right)^2 + \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right)^2 + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j Pe(t) + \alpha \beta^2 j^2 Pe(t)^2 \right) \right)$$

3. On dispose de tous les éléments pour appeler à la rescousse la commande **EulerLagrange** du paquetage **VariationalCalculus**.

$$> \mathbf{with(VariationalCalculus):}$$

$$\mathbf{sol := EulerLagrange(f, t, Pe(t));}$$

$$\mathbf{eulag := sol[1];}$$

$$sol := \left\{ \frac{e^{-\rho t} \left(2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j + 2 \alpha \beta^2 j^2 Pe(t) \right)}{j^2 \beta^2} \right.$$

$$+ \frac{\rho e^{-\rho t} \left(2 \left(\frac{d}{dt} Pe(t) \right) + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) + 2 \alpha \beta^2 j Pe(t) \right)}{j^2 \beta^2}$$

$$\left. - \frac{e^{-\rho t} \left(2 \left(\frac{d^2}{dt^2} Pe(t) \right) + 2 \alpha \beta^2 \left(\frac{d^2}{dt^2} Pe(t) \right) + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j \right)}{j^2 \beta^2} \right\}$$

$$\tag{7.6}$$

$$\begin{aligned}
eulag := & \frac{e^{-\rho t} \left(2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j + 2 \alpha \beta^2 j^2 Pe(t) \right)}{j^2 \beta^2} \\
& + \frac{\rho e^{-\rho t} \left(2 \left(\frac{d}{dt} Pe(t) \right) + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) + 2 \alpha \beta^2 j Pe(t) \right)}{j^2 \beta^2} \\
& - \frac{e^{-\rho t} \left(2 \left(\frac{d^2}{dt^2} Pe(t) \right) + 2 \alpha \beta^2 \left(\frac{d^2}{dt^2} Pe(t) \right) + 2 \alpha \beta^2 \left(\frac{d}{dt} Pe(t) \right) j \right)}{j^2 \beta^2}
\end{aligned} \tag{7.6}$$

La résolution de cette équation différentielle linéaire homogène du second ordre à coefficients constants ne pose aucun problème.

> dsolve(eulag, Pe(t));

$$\begin{aligned}
Pe(t) = & _C1 e^{\frac{1}{2} \frac{\left(\rho + \rho \alpha \beta^2 + \sqrt{\rho^2 + 2\rho^2 \alpha \beta^2 + \rho^2 \alpha^2 \beta^4 + 4\alpha \beta^2 j^2 + 4\alpha^2 \beta^4 j^2 + 4\rho \alpha \beta^2 j + 4\rho \alpha^2 \beta^4 j} \right) t}{1 + \alpha \beta^2}} \\
& + _C2 e^{-\frac{1}{2} \frac{\left(-\rho - \rho \alpha \beta^2 + \sqrt{\rho^2 + 2\rho^2 \alpha \beta^2 + \rho^2 \alpha^2 \beta^4 + 4\alpha \beta^2 j^2 + 4\alpha^2 \beta^4 j^2 + 4\rho \alpha \beta^2 j + 4\rho \alpha^2 \beta^4 j} \right) t}{1 + \alpha \beta^2}}
\end{aligned} \tag{7.7}$$

4. Après injection des valeurs fournies pour les paramètres, on récupère l'équation d'Euler-Lagrange que l'on résout en tenant compte des conditions aux bornes.

> rho, alpha, beta, j := 0.05, 1, 0.5, 0.2;

eulag;

Pestars := dsolve({eulag, Pe(0)=15, Pe(40)=0}, Pe(t));

rho, alpha, beta, j := 0.05, 1, 0.5, 0.2

$$\begin{aligned}
& 100.0000000 e^{-0.05t} \left(0.100 \left(\frac{d}{dt} Pe(t) \right) + 0.0200 Pe(t) \right) \\
& + 5.000000000 e^{-0.05t} \left(2.50 \left(\frac{d}{dt} Pe(t) \right) + 0.100 Pe(t) \right) \\
& - 100.0000000 e^{-0.05t} \left(2.50 \left(\frac{d^2}{dt^2} Pe(t) \right) + 0.100 \left(\frac{d}{dt} Pe(t) \right) \right) \\
Pestars := Pe(t) = & - \frac{15 e^{1 - \sqrt{17}} e^{\frac{1}{40} (1 + \sqrt{17}) t}}{e^{1 + \sqrt{17}} - e^{1 - \sqrt{17}}} + \frac{15 e^{1 + \sqrt{17}} e^{-\frac{1}{40} (-1 + \sqrt{17}) t}}{e^{1 + \sqrt{17}} - e^{1 - \sqrt{17}}}
\end{aligned} \tag{7.8}$$

Il ne reste plus qu'à vérifier graphiquement que les anticipations de prix décroissent continûment sur la période pour s'annuler à la date terminale.

> plot(rhs(Pestars), t=0..40);

